

Using SAS To Improve Efficiency In Report Production and Dissemination

Dragos Daniel Capan DCapan@cihi.ca

William Lu WLu@cihi.ca

What is HSMR

- HSMR stands for Hospital Standardized Mortality Ratio
- Initially calculated at hospital level
- Now calculated at organization and region level as well
- 291 registered qualifying hospitals
- 101 regions & organizations
- 554 reports to be sent



Dissemination – the old way


- All reportable hospitals pdfs in one folder
- Create chunks of about 25 files to send to printer
- Print labels, letter, toolkit (FAQ, Tech notes...)
- Use Canada Post to send the packages



Hospital Standardized Mortality Ratio (HSMR) - FY 2007/08 Q3 Quarterly Report
CIHI HSMR Methodology version 3.0

Institution name

HSMR all cases



| Fiscal Quarter | Cases | Deaths | HSMR | 95% CI | Peer Range** | Peer QR** |
|-----------------------|--------------|------------|------------|------------------|-----------------|------------------|
| Q1 Apr-Jun 2007 | 1,188 | 82 | 104 | 82 - 129 | | |
| Q2 Jul-Sep 2007 | 1,126 | 92 | 111 | 90 - 137 | | |
| Q3 Oct-Dec 2007 | 1,150 | 86 | 114 | 92 - 141 | | |
| Q4 Jan-Mar 2008 | - | - | - | - | | |
| YTD (FY 07/08) | 3,464 | 260 | 112 | 105 - 120 | 79 - 143 | 104 - 135 |

HSMR excluding palliative care

| Fiscal Quarter | Cases | Deaths | HSMR | 95% CI | Peer Range** | Peer QR** |
|-----------------------|--------------|------------|-----------|-----------------|-----------------|-----------------|
| Q1 Apr-Jun 2007 | 1,167 | 73 | 95 | 78 - 110 | | |
| Q2 Jul-Sep 2007 | 1,110 | 85 | 103 | 90 - 114 | | |
| Q3 Oct-Dec 2007 | 1,153 | 70 | 93 | 74 - 107 | | |
| Q4 Jan-Mar 2008 | - | - | - | - | | |
| YTD (FY 07/08) | 3,430 | 228 | 98 | 90 - 108 | 55 - 140 | 75 - 121 |

CONFIDENTIAL PRELIMINARY DATA

ID:12345, Peer Group 4, Organization name

Data submitted as of 07-Mar-08

Periods Outstanding: 0

Report created by CIHI on 21-Apr-08



ODS pdf basics

```
ods listing close;
ods pdf body="PATH\file_name.pdf"
style=styles.fixeddocPrinter2 notoc ;
```

```
/* SAS code */
/* Proc print data=try; run; */
```

```
ods pdf close;
```

```
ods listing;
```

**A preset style
or your own
style**




ODS LAYOUT

```
ods listing close;  
ods pdf body="PATH\file_name.pdf"  
style=styles.fixeddocPrinter2 notoc ;
```

```
ods layout start /*options*/;  
ods region ;
```

```
/*SAS code */
```

```
ods layout end;
```

```
ods pdf close;  
ods listing;
```



ODS LAYOUT

- Absolute layout
 - allows more control on the output
 - difficult to use
 - size of each region is fixed when you specify it

```
ods layout start width=8in height=12in;
ods region width=6in height=2in x=0 y=0;
proc print;
run;

ods layout end;
```



ODS LAYOUT

- Gridded layout
 - not that much control
 - easier to use than absolute layout
 - The region tries to accommodate the output

```
ods layout start rows= 5 columns=2 width=8in height=12in;
ods region ;
ods pdf text="place your text here";
ods region;
proc report data=try;
run;
ods layout end;
```



Make the output look nice

- Proc template
 - Default style for ODS pdf is *printer*
 - Create your own style from scratch or modify one of the existing styles
- Proc report
 - A lot more flexibility and options than proc print
 - Use *call define* to change rows or columns that meet some criterion
- Escape character
 - Used for formatting



Dissemination – the better way

- Create folders for each contact name (corporation / region)
- Generate unique password for each folder
- Password protect the pdfs and send them to the appropriate folder
- Use SAS to send the reports by email (all pdfs in a folder get attached to one email)



The 'magic' file

| ID | Folder | Email | Multi | password |
|-------|---------------|----------------|-------|----------|
| 12345 | Hospital A | A@x.ca | 0 | Gtfr4yt |
| 12345 | Region A | B@y.ca | 1 | gHyt45k |
| 1111 | Corporation B | H@z.ca | 0 | Abu7t |
| 11112 | Corporation B | H@z.ca | 0 | Abu7t |
| 11113 | Corporation B | H@z.ca | 0 | Abu7t |
| 11112 | Region B | Lhin@on .ca | 1 | Jyt5Re |



Generating passwords

```
%let text=q w e r t y u o p a s d f g h j k z x c v
      b n m a b c q a z w s x e d c r f v t g b y h n u
      j m k o p a s d f g h j k w r y p o u t e i;
```

```
%macro pass(in=, out=, seedm=1);
  data pass;
  set &in end=last;
  %do i=1 %to 7;
    l&i=int(ranuni(%eval(&i*&seedm.))*27)+1;
  %end;
  indx+1;
  if last then call symput('nr', indx);
run;
```



Generating passwords

```
%do i=1 %to &nr;
  data pass_&i;
  set pass;
  if indx=&i;
  %do j=1 %to 7;
  call symputx("v&j",left(l&j));
  %end;
run;
```

```
%do j=1 %to 6;
  %let m=%eval(&j+1);
  %if %sysevalf(&&v&j/&&v&m)>1 %then
    %let l&j=%upcase(%scan(&text,&&v&j));
  %else %let l&j=%scan(&text,&&v&j);
%end;
```



Generating passwords

```
%let password=&l1.&l2.&l3.&l4.&l5.&v6.&l6;
```

```
data pass_&i; set pass_&i; length password $ 8.;  
password="&password"; run;  
%end;
```

```
data &out; set %do i=1 %to &nr; pass_&i %end;;  
drop indx 1: ; run;
```

```
proc datasets; delete %do i=1 %to &nr; pass_&i  
%end;; run;
```

```
%mend pass;
```



Generating folders

- 170 folders to be created
- Generate a batch file using SAS and the 'magic' file
- Run the batch file either in SAS or by double clicking it

To create a folder using the command line:

```
md "C:/My Documents/ HUG"
```



Generating folders

```

data _null_;
  file "&path_to_batch\folders.bat";
  set folders;
  length cmd $256.;
  cmd='md" ' || "&folder\" || trim(left(folder_name)) || '"';
  put cmd;
run;

```

```

options XMIN NOXWAIT XSYNC;

```

```

data _null_;
  call system( '"&path_to_batch\folders.bat" ');
run;

```



Password protect the files

- Starting with version 9.2 SAS can add a password to a pdf file
- If you don't have SAS 9.2 then you can use other software
- Ideally, it should support command line so that you can automate the process (run the software using SAS)
- PDFTK allows you to merge, split, encrypt, apply watermark, attach files to the pdf document,...



Password protect the files

```
%encrypt_pdf( INPUT_PDF="path1\input.pdf" ,  
OUTPUT_PDF="path2\output.pdf" ,  
USER_PW="abcd123" )
```

OR

```
%SYSEXEC "path\pdftk.exe" "path_to\input.pdf"  
OUTPUT "path_to\output.pdf"  
ALLOW PRINTING COPYCONTENTS  
ENCRYPT_128BIT USER_PW "abcd123" ;
```



Re-protecting the files

- There were 554 files to be protected and about 25 were not
- Wrote a macro that goes through all the folders, checks if the protected files exists, and if not, runs the `%encrypt_pdf` again

```
%let check=%sysfunc(fileexist("&folderm\
&folder\HSMR_FY07_Q3_&ID._protected.pdf"));
%if &check=0 %then %do;
    %encrypt_pdf(...)
%end;
```



Emailing the reports with SAS

- Using the 'magic' file, subset the records that belong to each folder
- Create macro variables that record:
 - IDs (used to make the name of the file) in that folder
 - Folder name
 - Email address
 - Number of reports



Emailing the reports with SAS

```
%do i=1 %to &rep;
```

```
data _null_;
```

```
set reportable_email(where =(indx_email=&i)) end=last;
```

```
call symputx('ID' || left(put(_N_,2.)),ID);
```

```
if last then do;
```

```
    call symputx('folder',folder_name);
```

```
    call symputx('nr',_N_);
```

```
    call symputx('email', contact_email);
```

```
end;
```

```
run;
```

```
%do j=1 %to &nr;
```

```
%let attch&j.="&fm.\&folder.\HSMRFY07_&&ID&j.._pr.pdf";
```

```
%end;
```



Emailing the reports with SAS

```
filename outbox email "null" emailsys=mapi  
  from="HSMR" attach=(%do j=1 %to &nr; &&attach&j  
  %end; &faq &tech &attachword);
```

```
data _null_;  
  file outbox  
    to="&email" cc="&email2"  
  subject="HSMR Q3 - &folder ";  
  put 'Text1';  
  put ' ';  
  put 'Text2';  
  put ' ';  
run;  
end;
```



Email and Encrypting PDF file

- Email in SAS
 - 3 types
 - SMTP options
 - Statements
- Encrypting PDF file
 - In SAS
 - By pdftk



Email in SAS – 3 types

- **MAPI**

Mail API, such as Microsoft Exchange, Lotus Notes Version 5, and Eudora

- **VIM**

Vendor Independent Mail, such as Lotus cc:Mail and Lotus Notes Version 4

- **SMTP**

Simple Mail Transfer Protocol



Email in SAS – SMTP options

- EMAILSYS
 - SMTP
- EMAILHOST
 - specifies the SMTP server that supports e-mail access
- EMAILPORT
 - specifies the port to which the SMTP server is attached
- EMAILID
 - provides the username
- EMAILPW
 - provides the password



Email in SAS – Statements

```
filename outbox email from="Abc@fake.com"
```

```
    to="Cfg@fake.com"
```

```
    subject="HSMR demo"
```

```
    attach='C:\HSMR\HMSR.log';
```

```
data _null_;
```

```
    file outbox;
```

```
    put 'Hi Daniel,';
```

```
    put 'This is demo only.';
```

```
    put ' ';
```

```
    put 'William';
```

```
run;
```



Email in SAS – Document

- SAS OnlineDOC

<http://support.sas.com/onlinedoc/913/getDoc/en/lrdict.hlp/a002058232.htm>



Encrypting PDF file – in SAS

- Create PDF by ODS
- Encrypting PDF in SAS

Options: PDFSECURITY and PDFPASSWORD

(<http://support.sas.com/documentation/cdl/en/lrdict/59540/HTML/default/a003086061.htm>

<http://support.sas.com/documentation/cdl/en/lrdict/59540/HTML/default/a003080949.htm>)

PDFSECURITY= HIGH | LOW | NONE

PDFPASSWORD=(OWNER="password", OPEN="password")

(only available from SAS 9.2)



Encrypting PDF file – by pdftk

<http://www.accesspdf.com/pdftk/>

Pdftk - the PDF Toolkit



If PDF is electronic paper, then pdftk is an electronic staple-remover, hole-punch, binder, secret-decoding, and X-Ray-glasses. Pdftk is a command-line tool for doing everyday things with PDF documents. Keep one in the top drawer of your desktop and use it to:

- Merge PDF Documents
- Split PDF Pages into a New Document
- Decrypt Input as Necessary (Password Required)
- Encrypt Output as Desired
- Fill PDF Forms with FDF Data and/or Flatten Forms
- Apply a Background Watermark
- Report on PDF Metrics such as Metadata, Bookmarks, and Page Labels
- Update PDF Metadata
- Attach Files to PDF Pages or the PDF Document
- Unpack PDF Attachments
- Burst a PDF Document into Single Pages
- Uncompress and Re-Compress Page Streams
- Repair Corrupted PDF (Where Possible)

Pdftk allows you to manipulate PDF easily and freely. It does not require Acrobat, and it runs on Windows, Linux, Mac OS X, FreeBSD and Solaris.

Pdftk is free software (GPL).

Encrypting PDF file – by pdftk

- Usage (sample)

```
C:\pdftk.exe "c:\HSMR\HSMR_raw.pdf"  
  output "c:\HSMR\HSMR_encrypted.pdf"  
  encrypt_128bit  
  user_pw "ABCD"  
  allow printing CopyContent
```

```
--PDF file will be encrypted  
--encrypted PDF file  
--similar to option PDFSECURITY  
--similar to option PDFPASSWORD  
--control the encrypted PDF file,  
--allow to be printed and copy/paste
```



Encrypting PDF file – by pdftk

- SAS macro

```
%macro encrypt_pdf( input_pdf,  
                    output_pdf,  
                    user_pw,  
                    security          =HIGH,  
                    pgm                ="c:\pdftk.exe");  
  
%if %upcase(&security) =LOW %then %let encrypt_bit=encrypt_40bit;  
%if %upcase(&security) =HIGH %then %let encrypt_bit=encrypt_128bit;  
  
%sysexec &pgm &input_pdf output &output_pdf &encrypt_bit user_pw &user_pw allow  
printing CopyContents;  
%mend encrypt_pdf;
```

*/*example*

```
%encrypt_pdf(INPUT_PDF="Path\input.pdf",  
              OUTPUT_PDF="Path\output.pdf",  
              USER_PW="abcd123");*/
```



Some references

- **ODS LAYOUT: Arranging ODS Output as You See Fit** (*Brian T. Schellenberger, SAS Institute Inc., Cary, NC*)
- **Presentation-Quality Tabular Output via ODS** (*Brian T. Schellenberger*)
- **Funny ^Stuff~ in My Code: Using ODS ESCAPECHAR** (*Cynthia L. Zender, SAS Institute Inc., Cary, NC – SAS Global Forum 2007*)

